
Writing a Custom BasicEffect 

The Need for our own Basic Effect 
Previously, our need for our own custom effect was related to the lack of support for certain features of 

the built in BasicEffect class on the Xbox 360. From the looks of the updated API though, these problems 

have since been remedied, so our needs must be updated to show that we should still create our own 

custom basic effect. My major reason for still including this tutorial is the general fact that learning 

shaders is hard, and that a tutorial series on XNA should include as many components of XNA as it can. 

So, without further ado, here comes a learning experience in HLSL and XNA shaders! 

[To make it a bit easier to see what is supposed to be going on, I will include a picture of my demo after each section of the shader is added so 

everyone can be sure they are getting correct results.] 

A Quick Vector Primer 
Since lighting and pretty much any other shaders use a lot of vector math I thought I’d explain in general 

the functions I will use here. Hopefully this will help people less familiar with vector math to get through 

this tutorial without too much fuss. 

A dot product of two vectors is defined like this: 

u.v = ||u|| * ||v|| * cos(theta) 

This basically means that the dot product equals the length of the two vectors times the angle between 

them. Now, we are most interested in the angle between them, and so taking advantage of the fact that 

we have normalized all of our vectors, the lengths are always 1. This makes the equation look like this: 

u.v = 1 * 1 * cos(theta) 

Basically, this allows us to find a reference to the angle between two vectors just using the dot product 

which just happens to be a built in shader function. 

Since our lighting will need to be an amount between 0.0 and 1.0 (a basic brightness scale), using the 

already given cosine of the angle will be perfect. This is already the percent of light that would be 

received at that angle and can be used directly. The second two components of our lighting model 

included here (a basic Phong model) rely heavily on this fact. With this simple calculation, we can create 

quite a few lighting effects with ease. 

To begin our process, create a new file in the HMContent folder called BasicShader.fx. 

Starting with a Simple Transform 
For starters, the basic shader will need to be able to transform any model we 

have on screen. This is a fairly simple process that requires very few additions 

to the code we already have in place for our shaders. Since this is going to be a 

fairly complex shader very soon, I am going to start from scratch so everything 

is very well explained (I hope.)  The shader for transforming our model will 

start out with exactly the same code that our previous TransformColorTexture 

shader did (so I guess we aren’t really starting from scratch after all.) 

All we need to do is just copy paste the code from TransformColorTexture.fx for now and we can get to 

adding on to it for some shader based lighting. 

Since we are going to be playing pretty deep into things with the shaders in this tutorial, I am going to be 

using a teapot mesh so we can see a lot of nice curves and edges to help us get a really good idea of 

what is actually going on with everything. Here are my changes to HMDemo for reference: 

private static readonly HMShader shader2 = new HMShader(@"HMContent/Shaders/BasicShader"); 

 

// In HMDemo.Main() 

HMEffectManager.AddEffect("BS", shader2); 

 

model.Shader = "BS"; 

Adding in some Ambient Light 
The best way I can think of describing ambient lighting (which of course 

involves snow, because it is minus 8 Fahrenheit outside right now) is to 

imagine a pitch black night in the middle of winter. Everything outside is still 

very visible because of the HUGE amount of scattering of any small lights 

around due to all the snow. This is referred to as ambient lighting. Ambient 

light is a very simple calculation that makes all parts of a mesh the same color 

(although I do make it a bit more complicated in a second.) To use ambient 

light, we will add a constant to the top of our shader called “AmbientColor” (keeping close to BasicEffect 

parameters.) This will be a float3 that will store a color to base everything in the scene by. This color will 

only be used by the pixel shader to calculate final pixel color. Here is the changed function for that (I 

renamed it from ColorTexture to BasicShader as well): 

float3 AmbientColor; 

 

float4 BasicShader(VS_INPUT Input) : COLOR0 { 

    float4 TextureColor = Input.Color.rgba * tex2D(TextureSampler, Input.Texcoord); 

    float4 LightingColor = float4(AmbientColor, 1.0f); 

    return TextureColor > 0 ? TextureColor * LightingColor : LightingColor; 

}; 

We are using the LightingColor variable because it will have a lot of other components of the final light 

calculation added to it before we are finished. This addition will also require being able to set the 

AmbientColor from the engine itself, and since each object could potentially have its own ambient color 

we will make some code additions to our model object to deal with these situations. Here are the 

additions to the HMModel class we will need for the ambient and other components later on: 

using HMEngine.HMEffects; 

 

public Vector3 AmbientColor { get; set; } 

public Vector3 DiffuseColor { get; set; } 

public float SpecularPower { get; set; } 

 

public HMModel(string asset) { 

    myAsset = asset; 

    Scaling = new Vector3(1); 

 

    AmbientColor = new Vector3(0.15f); 

    DiffuseColor = new Vector3(0.25f); 

    SpecularPower = 8; 

} 

 

internal void SetMaterialProperties() { 

    Effect effect = HMEffectManager.ActiveShader.Effect; 

 

    if (null != effect.Parameters["AmbientColor"]) { 

        effect.Parameters["AmbientColor"].SetValue(AmbientColor); 

    } 

    if (null != effect.Parameters["DiffuseColor"]) { 

        effect.Parameters["DiffuseColor"].SetValue(DiffuseColor); 

    } 

    if (null != effect.Parameters["SpecularPower"]) { 

        effect.Parameters["SpecularPower"].SetValue(SpecularPower); 

    } 

 

    effect.CommitChanges(); 

} 

 

public override void Render(GraphicsDevice myDevice) { 

    // Matrix Code 

 

    foreach (ModelMesh mesh in myModel.Meshes) { 

        // Effect World Code 

 

        foreach (ModelMeshPart part in mesh.MeshParts) { 

            // Device Vertex Code 

            myDevice.Textures[0] = ((BasicEffect)part.Effect).Texture; 

 

            // Set up the shader properties for this mesh part 

            SetMaterialProperties(); 

 

            // Draw Code 

        } 

    } 

} 

The SetMaterialProperties function will simply grab the current effect for the object and send in the 

values if needed. I have included a default value for ambient lighting in case we forget to explicitly call 

the property for setting that value some time. We also added a call to SetMaterialProperties in the 

Render function right before the draw code so we can make sure each object is drawn with its settings. 

Running the application now will really only show us a gray teapot instead of a black one, which isn’t 

very impressive. This boring gray mesh will simply not do for ambient lighting (at least in my book), so I 

am going to add in a component to help us see things a bit better for what they are. This will be called 

the edge component. 

Adding the Edge Component 
An edge component is basically a simple number that describes the difference 

between the angle of a surface, and the direction that an object is being 

looked at from. To accomplish the addition of this component, we will need to 

send in the camera’s current position to the shader so we can use it to find the 

view direction. In the shader, this will be called the “EyePosition” (keeping 

with the naming of parameters in Microsoft’s BasicEffect). We will send this in 

the SetParameters function of HMShader like so: 

if (null != Effect.Parameters["EyePosition"]) { 

    Effect.Parameters["EyePosition"].SetValue(HMCameraManager.ActiveCamera.Position); 

} 

The shader itself will also use some normals calculation, but these are already being sent in by the 

VertexBuffer in the Render functions. To start the edge component calculation, we first need to 

translate the normal and create a ViewDirection for ourselves in the vertex shader and pass them into 

the pixel shader: 

float3 EyePosition; 

 

struct VS_INPUT { 

    float4 Position         : POSITION0;  

    float2 Texcoord         : TEXCOORD0;  

    float4 Color            : COLOR0; 

    float3 Normal           : NORMAL0; 

}; 

 

struct VS_OUTPUT { 

    float4 Position         : POSITION0; 

    float2 Texcoord         : TEXCOORD0; 

    float4 Color            : COLOR0; 

    float3 Normal           : TEXCOORD1; 

    float3 ViewDirection    : TEXCOORD2; 

}; 

 

VS_OUTPUT Transform(VS_INPUT Input) {  

    VS_OUTPUT Output;  

  

    float4 worldPosition = mul(Input.Position, World); 

    float4 viewPosition = mul(worldPosition, View); 

     

    Output.Position = mul(viewPosition, Projection); 

    Output.Texcoord = Input.Texcoord; 

    Output.Color = Input.Color; 

    Output.Normal = mul(Input.Normal, World); 

    Output.ViewDirection = EyePosition - worldPosition; 

  

    return Output;  

} 

 

struct PS_INPUT { 

    float2 Texcoord         : TEXCOORD0; 

    float4 Color            : COLOR0; 

    float3 Normal           : TEXCOORD1; 

    float3 ViewDirection    : TEXCOORD2; 

}; 

 

float4 BasicShader(PS_INPUT Input) : COLOR0 { 

    float3 Normal = normalize(Input.Normal); 

    float3 ViewDirection = normalize(Input.ViewDirection); 

     



P a g e  | 5 

Tutorial 7 - Writing a Custom BasicEffect 

    float EdgeComponent = dot(Normal, ViewDirection); 

    float3 TotalAmbient = saturate(AmbientColor * EdgeComponent); 

 

    float4 TextureColor = Input.Color.rgba * tex2D(TextureSampler, Input.Texcoord); 

    float4 LightingColor = float4(TotalAmbient, 1.0f); 

     

    return TextureColor > 0 ? TextureColor * LightingColor : LightingColor; 

}  

We started by getting the object’s position (after world transformation of course) and used that with the 

eye position to get a direction from the camera to the object. We can use this new direction vector and 

the passed through normal to find out whether a vertex is facing the camera or not in our pixel shader. 

Finally, at the end of the pixelshader, we make sure our texture color is greater than zero (since it will 

destroy the lighting value in the multiplication if it is) before choosing to return it and our lighting value 

or just the lighting. This will be even more specific later on when only certain parts of a mesh have 

certain types of lighting. 

Adding the Diffuse Lighting  
Diffuse lighting is caused by directional lights hitting one side of an object but 

not the other. This is a fairly easy to visualize type of light (especially with all 

my lovely pictures), so I won’t go into detail describing it. Most diffuse lighting 

is calculated with an essentially infinite distance light source (something like 

the sun), so I am going to pass in a light direction and not a position for now. 

We are also going to need to pass in a color for the light. These will be passed 

in the HMShader class and not the material section because we will have 

multiple lights at some point and materials are only set once, whereas these properties can be set later 

in a loop. Here are the two new SetParameters parts for passing these: 

if (null != Effect.Parameters["LightDirection"]) { 

    Effect.Parameters["LightDirection"].SetValue(new Vector3(1, -1, -1)); 

} 

if (null != Effect.Parameters["LightDiffuseColor"]) { 

    Effect.Parameters["LightDiffuseColor"].SetValue(new Vector3(0.5f)); 

} 

The light direction here is at 45 degrees in all directions down, away, and to the right. These values will 

be changed by light objects later on, but for now we will just use settings. Not only is the light color 

important to us, but different objects actually reflect this light in different ways based on their colors. 

The DiffuseColor we added in SetMaterialProperties earlier will handle passing this value to the shader 

for us so we can start adding them into the shader and do its calculations as well. Diffuse light relies on 

the angle between the direction to the light and the surface normal. We already have the surface parts 

figured out for the edge component, so we just need to pass the needed values on to the pixel shader: 

float3 DiffuseColor; 

float3 LightDirection; 

float3 LightDiffuseColor; 

 

float4 BasicShader(PS_INPUT Input) : COLOR0 { 

    float3 Normal = normalize(Input.Normal); 

    float3 ViewDirection = normalize(Input.ViewDirection); 

    LightDirection = normalize(-LightDirection); 

    float EdgeComponent = dot(Normal, ViewDirection); 

    float3 TotalAmbient = saturate(AmbientColor * EdgeComponent); 

 

// Diffuse Calculations 

    float NDotL = dot(Normal, LightDirection); 

    float3 DiffuseAverage = (DiffuseColor + LightDiffuseColor) * 0.5f; 

    float3 TotalDiffuse = saturate(DiffuseAverage * NDotL); 

// End Diffuse 

 

    float4 TextureColor = Input.Color.rgba * tex2D(TextureSampler, Input.Texcoord); 

    float4 LightingColor = float4(TotalAmbient + TotalDiffuse, 1.0f); 

     

    return TextureColor > 0 ? TextureColor * LightingColor : LightingColor; 

} 

The diffuse part of the basic shader just takes the dot product of the normal and the vector from the 

surface to the light position (-LightDirection) and multiplies it by an average between the light’s and the 

object’s diffuse colors. Adding this to the ambient color in the final part and returning it out finishes off 

this part of the lighting. Hopefully you read the vector primer at the beginning of this tutorial, or you 

already know enough about vector and lighting calculations that this isn’t too complicated to look at. 

Adding the Specular Lighting 
For specular lighting, we need to once again add a couple of properties to the 

actual engine. First off we are going to want to add in a color for the specular 

component of the light. This is usually going to be the same as the diffuse 

color, but just in case we want to play with things later on we will add it as a 

separate color now. This goes in the SetParameters of HMShader with the 

other lighting stuff: 

if(null != Effect.Parameters["LightSpecularColor"]) { 

    Effect.Parameters["LightSpecularColor"].SetValue(new Vector3(0.5f)); 

} 

A simple color will not be enough though, as the calculations for how shiny an object is requires an 

exponent to be passed in as well. Since this is a per object property, we have put it in the 

SetMaterialProperties function of the HMModel class with a default value of 8. The shader calculations 

for the specular component of our lighting actually takes advantage of the NDotL that we have already 

calculated to find the reflection vector of the light in the direction it would be bounced back off of the 

object, and takes another dot product between this new reflection vector and the view to see how 

much of that light is being bounced to the viewer. Here are the shader additions for this: 

float SpecularPower; 

 

float3 LightSpecularColor; 

 

// In BasicShader() 

float3 Reflection       = normalize(2.0f * Normal * NDotL - LightDirection); 

float RDotV             = max(0.0f, dot(Reflection, ViewDirection)); 

float3 TotalSpecular    = saturate(LightSpecularColor * pow(RDotV, SpecularPower)); 

 

float4 LightingColor    = float4(TotalAmbient + TotalDiffuse + TotalSpecular, 1.0f); 

There you have it. We now have all of the basic lighting allowed by BasicEffect put together into our own 
shader that we are free to do with what we would like. There are still a few properties in the BasicEffect 
that we haven’t gotten to yet, but I will add those in another tutorial when we implement things like fog 
equations and other light types as well. For now, you should play around with different color and power 
properties of the light to get a better understanding of what everything in the shader is doing.  One 
other thing to make our lives a little easier from now on is to modify the HMEffectManager to use our 
new BasicShader as the default shader for objects in case we forget to set one. After adding this you can 
remove the second shader we added in the demo code earlier: 
 

internal static void SetActiveShader(string effectLabel) { 

    ActiveShader = null == effectLabel 

                       ? HMComponentManager.GetComponent<HMShader>("HMEngine.HMEffects.BasicShader") 

                       : HMComponentManager.GetComponent<HMShader>(effectLabel); 

} 

 

public override void Initialize() { 

    var BasicShader = new HMShader("HMContent/Shaders/BasicShader"); 

    AddEffect("HMEngine.HMEffects.BasicShader", BasicShader); 

 

    base.Initialize(); 

} 

Last but not least, here is the final full shader after putting this all together: 

float4x4 World;  

float4x4 View;  

float4x4 Projection; 

 

float3 EyePosition; 

 

float3 AmbientColor; 

float3 DiffuseColor; 

float  SpecularPower; 

 

float3 LightDirection; 

float3 LightDiffuseColor; 

float3 LightSpecularColor; 

  

sampler TextureSampler;  

  

struct VS_INPUT { 

    float4 Position         : POSITION0;  

    float2 Texcoord         : TEXCOORD0;  

    float4 Color            : COLOR0; 

    float3 Normal           : NORMAL0; 

}; 

 

struct VS_OUTPUT { 

    float4 Position         : POSITION0; 

    float2 Texcoord         : TEXCOORD0; 

    float4 Color            : COLOR0; 

    float3 Normal           : TEXCOORD1; 

    float3 ViewDirection    : TEXCOORD2; 

}; 

 

VS_OUTPUT Transform(VS_INPUT Input) {  

    VS_OUTPUT Output;  

  

    float4 worldPosition = mul(Input.Position, World); 

    float4 viewPosition = mul(worldPosition, View); 

     

    Output.Position = mul(viewPosition, Projection); 

    Output.Texcoord = Input.Texcoord; 

    Output.Color = Input.Color; 

    Output.Normal = mul(Input.Normal, World); 

    Output.ViewDirection = EyePosition - worldPosition; 

  

    return Output;  

} 

 

struct PS_INPUT { 

    float2 Texcoord         : TEXCOORD0; 

    float4 Color            : COLOR0; 

    float3 Normal           : TEXCOORD1; 

    float3 ViewDirection    : TEXCOORD2; 

}; 

  

float4 BasicShader(PS_INPUT Input) : COLOR0 { 

    float3 Normal = normalize(Input.Normal); 

    float3 ViewDirection = normalize(Input.ViewDirection); 

    LightDirection = normalize(-LightDirection); 

     

    float EdgeComponent = dot(Normal, ViewDirection); 

    float3 TotalAmbient = saturate(AmbientColor * EdgeComponent); 

 

    float NDotL = dot(Normal, LightDirection); 

    float3 DiffuseAverage = (DiffuseColor + LightDiffuseColor) * 0.5f; 

    float3 TotalDiffuse = saturate(DiffuseAverage * NDotL); 

     

    float3 Reflection = normalize(2.0f * Normal * NDotL - LightDirection); 

    float RDotV = max(0.0f, dot(Reflection, ViewDirection)); 

    float3 TotalSpecular = saturate(LightSpecularColor * pow(RDotV, SpecularPower)); 

 

    float4 TextureColor = Input.Color.rgba * tex2D(TextureSampler, Input.Texcoord); 

    float4 LightingColor = float4(TotalAmbient + TotalDiffuse + TotalSpecular, 1.0f); 

     

    return TextureColor > 0 ? TextureColor * LightingColor : LightingColor; 

}  

  

technique BasicShader {  

    pass P0 {  

        VertexShader = compile vs_2_0 Transform();  

        PixelShader = compile ps_2_0 BasicShader();  

    }  

} 

Screenshots 
For anyone who is looking to compare their output directly to my images, this is the keyboard handler I 

used to get the camera in the same place every time (just press R after the demo loads): 

private static void keyboard_OnKeyReleased(Collection<Keys> keys) { 

    if (keys.Contains(Keys.F)) { 

        game.ToggleFullScreen(); 

    } 

    if (keys.Contains(Keys.Escape)) { 

        game.Exit(); 

    } 

    if (keys.Contains(Keys.R)) { 

        HMCameraManager.ActiveCamera.Position = new Vector3(0, 0, 5); 

        HMCameraManager.ActiveCamera.Revolve(new Vector3(1, 0, 0), -60 * 0.01f); 

        HMCameraManager.ActiveCamera.Revolve(new Vector3(0, 1, 0), -90 * 0.01f); 

    } 

} 


